

ПРИМЕРЫ ФУНКЦИЙ И ИЛИ В EXCEL ДЛЯ ЗАПИСИ ЛОГИЧЕСКИХ ВЫРАЖЕНИЙ

Функция И в Excel принимает одно или несколько логических выражений и возвращает логическое значение, которое зависит от результата вычислений всех этих выражений. Если результат вычисления каждого выражения – значение ИСТИНА, функция И вернет истинное значение. Если результатом вычисления хотя бы одного из выражений является значение ЛОЖЬ, функция И вернет логическое ЛОЖЬ.

Функция ИЛИ выполняет проверку логических выражений, принятых в качестве аргументов, и возвращает значение ИСТИНА, если результат выполнения хотя бы одного из этих выражений является истинным.

ПРИМЕРЫ ИСПОЛЬЗОВАНИЯ ФУНКЦИЙ И ИЛИ В ЛОГИЧЕСКИХ ВЫРАЖЕНИЯХ EXCEL

Пример 1. В учебном заведении для принятия решения о выплате студенту стипендии используют два критерия:

1. Средний балл должен быть выше 4,0.
2. Минимальная оценка за профильный предмет – 4 балла.

Определить студентов, которым будет начислена стипендия.

Исходная таблица:

	A	B	C	D	E	F
1	Студенты	Предмет 1	Предмет 2	Предмет 3	Профильный	Стипендия
2	Студент 1	4	5	5	4	?
3	Студент 2	3	5	5	3	?
4	Студент 3	4	4	3	5	?
5	Студент 4	5	5	3	4	?
6	Студент 5	5	5	3	3	?
7	Студент 6	3	3	5	5	?
8	Студент 7	3	3	3	5	?
9	Студент 8	4	3	3	4	?

Для решения используем следующую формулу:

Аргументы функции

И

Логическое_значение1 E2>=4 = ИСТИНА

Логическое_значение2 СРЗНАЧ(B2:E2)>=4 = ИСТИНА

= ИСТИНА

Проверяет, все ли аргументы имеют значение ИСТИНА, и возвращает значение ИСТИНА, если истинны все аргументы.

Логическое_значение1: логическое_значение1;логическое_значение2;... от 1 до 255 проверяемых условий, которые могут принимать значение ИСТИНА либо ЛОЖЬ; они могут быть логическими значениями, массивами или

Значение: ИСТИНА

[Справка по этой функции](#)

=И(E2>=4;СРЗНАЧ(B2:E2)>=4)

Описание аргументов:

- $E2 \geq 4$ – проверка условия «является ли оценка за профильный предмет не менее 4 баллов?»;
- $CP3HACH(B2:E2) \geq 4$ – проверка условия «является ли среднее арифметическое равным или более 4?».

Запишем аналогичные формулы в остальных ячейках. В результате получим:

F2						
=И(E2>=4;CP3HACH(B2:E2)>=4)						
	A	B	C	D	E	F
1	Студенты	Предмет 1	Предмет 2	Предмет 3	Профильный	Стипендия
2	Студент 1	4	5	5	4	ИСТИНА
3	Студент 2	3	5	5	3	ЛОЖЬ
4	Студент 3	4	4	3	5	ИСТИНА
5	Студент 4	5	5	3	4	ИСТИНА
6	Студент 5	5	5	3	3	ЛОЖЬ
7	Студент 6	3	3	5	5	ИСТИНА
8	Студент 7	3	3	3	5	ЛОЖЬ
9	Студент 8	4	3	3	4	ЛОЖЬ

Как видно выше на рисунке студенты под номером: 2, 5, 7 и 8 - не получают стипендию.


ФОРМУЛА С ЛОГИЧЕСКИМ ВЫРАЖЕНИЕМ ДЛЯ ПОИСКА ИСТИННОГО ЗНАЧЕНИЯ

Пример 2. На предприятии начисляют премии тем работникам, которые проработали свыше 160 часов за 4 рабочих недели (задерживались на работе сверх нормы) или выходили на работу в выходные дни. Определить работников, которые будут премированы.

Исходные данные:

C3			
ЛОЖЬ			
	A	B	C
1	Премии		
2	Работник	Часы	В выходные
3	Работник 1	148	ЛОЖЬ
4	Работник 2	182	ЛОЖЬ
5	Работник 3	171	ИСТИНА
6	Работник 4	160	ИСТИНА
7	Работник 5	156	ИСТИНА
8	Работник 6	160	ЛОЖЬ
9	Работник 7	160	ЛОЖЬ
10	Работник 8	192	ЛОЖЬ
11	Работник 9	184	ИСТИНА
12	Работник 10	144	ЛОЖЬ

Формула для расчета:



=ИЛИ(B3>160;C3=ИСТИНА())

Описание аргументов:

- B3>160 – условие проверки количества отработанных часов;
- C3=ИСТИНА() – условие проверки наличия часов работы в выходные дни.

Используя аналогичную формулу проведем вычисления для всех работников. Полученный результат:

	A	B	C	D
1	Премии			
2	Работник	Часы	В выходные	Премия
3	Работник 1	148	ЛОЖЬ	ЛОЖЬ
4	Работник 2	182	ЛОЖЬ	ИСТИНА
5	Работник 3	171	ИСТИНА	ИСТИНА
6	Работник 4	160	ИСТИНА	ИСТИНА
7	Работник 5	156	ИСТИНА	ИСТИНА
8	Работник 6	160	ЛОЖЬ	ЛОЖЬ
9	Работник 7	160	ЛОЖЬ	ЛОЖЬ
10	Работник 8	192	ЛОЖЬ	ИСТИНА
11	Работник 9	184	ИСТИНА	ИСТИНА
12	Работник 10	144	ЛОЖЬ	ЛОЖЬ

То есть, оба условия возвращают значения ЛОЖЬ только в 4 случаях.

ФОРМУЛА С ИСПОЛЬЗОВАНИЕМ ЛОГИЧЕСКИХ ФУНКЦИЙ ИЛИ И ЕСЛИ

Пример 3. В интернет-магазине выполнена сортировка ПК по категориям «игровой» и «офисный». Чтобы компьютер попал в категорию «игровой», он должен соответствовать следующим критериям:

1. CPU с 4 ядрами или таковой частотой не менее 3,7 ГГц;
2. ОЗУ – не менее 16 Гб
3. Видеоадаптер 512-бит или с объемом памяти не менее 2 Гб

Распределить имеющиеся ПК по категориям.

Исходные данные:

	A	B	C	D	E	F	G
1	Компьютеры						
2	ПК	Число ядер CPU	Тактовая частота	ОЗУ, Гб	Видеоадаптер, Гб	Видеоадаптер, разрядность	Категория
3	ПК1	8	3,9	8	1	256	?
4	ПК2	4	3	16	2	512	?
5	ПК3	2	3,2	12	4	512	?
6	ПК4	4	3,7	32	2	128	?
7	ПК5	6	4,2	6	1	128	?
8	ПК6	2	4	4	1	512	?
9	ПК7	2	4,7	12	2	512	?
10	ПК8	4	3	16	4	256	?
11	ПК9	4	2,7	32	2	512	?

Для определения категории используем следующую формулу:

Аргументы функции

ЕСЛИ

Лог_выражение: И(ИЛИ(В3 >= 4; С3 >= 3,7); D3 >= 16; ИЛИ(Е3 >= 2; F3 >= 512)) = ЛОЖЬ

Значение_если_истина: "Игровой" = "Игровой"

Значение_если_ложь: "Офисный" = "Офисный"

Проверяет, выполняется ли условие, и возвращает одно значение, если оно выполняется, и другое значение, если нет.

Лог_выражение: любое значение или выражение, которое при вычислении дает значение ИСТИНА или ЛОЖЬ.

Значение: Офисный

[Справка по этой функции](#) [OK] [Отмена]

Описание аргументов:

И(ИЛИ(В3>=4;С3>=3,7);D3>=16;ИЛИ(Е3>=2;F3>=512)) – функция И, проверяющая 3 условия:

1. Если число ядер CPU больше или равно 4 или тактовая частота больше или равна 3,7, результат вычисления ИЛИ(В3>=4;С3>=3,7) будет значение истинное значение;
2. Если объем ОЗУ больше или равен 16, результат вычисления D3>=16 – ИСТИНА;
3. Если объем памяти видеоадаптера больше или равен 2 или его разрядность больше или равна 512 – ИСТИНА.

«Игровой» - значение, если функция И возвращает истинное значение;

«Офисный» - значение, если результат вычислений И – ЛОЖЬ.

Этапы вычислений для ПК1:

1. =ЕСЛИ(И(ИЛИ(ИСТИНА;ИСТИНА);ЛОЖЬ;ИЛИ(ЛОЖЬ;ЛОЖЬ));"Игровой";"Офисный")
2. =ЕСЛИ(И(ИСТИНА;ЛОЖЬ;ЛОЖЬ);"Игровой";"Офисный")

3. =ЕСЛИ(ЛОЖЬ;"Игровой";"Офисный")

4. Возвращаемое значение – «Офисный»

Результаты расчетов для всех ПК:

G3								=ЕСЛИ(И(ИЛИ(B3>=4;C3>=3,7);D3>=16;ИЛИ(E3>=2;F3>=512));"Игровой";"Офисный")		
	A	B	C	D	E	F	G	H	I	
1	Компьютеры									
2	ПК	Число ядер CPU	Тактовая частота	ОЗУ, ГБ	Видеоадаптер, ГБ	Видеоадаптер, разрядность	Категория			
3	ПК1	8	3,9	8	1	256	Офисный			
4	ПК2	4	3	16	2	512	Игровой			
5	ПК3	2	3,2	12	4	512	Офисный			
6	ПК4	4	3,7	32	2	128	Игровой			
7	ПК5	6	4,2	6	1	128	Офисный			
8	ПК6	2	4	4	1	512	Офисный			
9	ПК7	2	4,7	12	2	512	Офисный			
10	ПК8	4	3	16	4	256	Игровой			
11	ПК9	4	2,7	32	2	512	Игровой			

ОСОБЕННОСТИ ИСПОЛЬЗОВАНИЯ ФУНКЦИЙ И, ИЛИ В EXCEL

Функция И имеет следующую синтаксическую запись:

=И(логическое_значение1;[логическое_значение2];...)

Функция ИЛИ имеет следующую синтаксическую запись:

=ИЛИ(логическое_значение1;[логическое_значение2];...)

Описание аргументов обеих функций:

- **логическое_значение1** – обязательный аргумент, характеризующий первое проверяемое логическое условие;
- [логическое_значение2] – необязательный второй аргумент, характеризующий второе (и последующие) условие проверки.

Сравнения результатов, возвращаемых функциями И, ИЛИ в однотипных ситуациях:

	A	B	C	D
	Результат проверки первого условия	Результат проверки второго условия	Значение, возвращаемое функцией И	Значение, возвращаемое функцией ИЛИ
1	ЛОЖЬ	ЛОЖЬ	ЛОЖЬ	ЛОЖЬ
2	ЛОЖЬ	ИСТИНА	ЛОЖЬ	ИСТИНА
3	ИСТИНА	ЛОЖЬ	ЛОЖЬ	ИСТИНА
4	ИСТИНА	ИСТИНА	ИСТИНА	ИСТИНА

Примечания 1:

1. Аргументами должны являться выражения, результатами вычисления которых являются логические значения.
2. Результатом вычислений функций И/ИЛИ является код ошибки #ЗНАЧ!, если в качестве аргументов были переданы выражения, не возвращающие в результате вычислений числовые значения 1, 0, а также логические ИСТИНА, ЛОЖЬ.

Примечания 2:

1. Функция И чаще всего используется для расширения возможностей других логических функций (например, ЕСЛИ). В данном случае функция И позволяет выполнять проверку сразу нескольких условий.
2. Функция И может быть использована в качестве формулы массива. Для проверки вектора данных на единственное условие необходимо выбрать требуемое количество пустых ячеек, ввести функцию И, в качестве аргумента передать диапазон ячеек и указать условие проверки (например, =И(A1:A10>=0)) и использовать комбинацию Ctrl+Shift+Enter. Также может быть организовано попарное сравнение (например, =И(A1:A10>B1:B10)).
3. Как известно, логическое ИСТИНА соответствует числовому значению 1, а ЛОЖЬ – 0. Исходя из этих свойств выражения $=(5>2)*(2>7)$ и $=И(5>2;2>7)$ эквивалентны (Excel автоматически выполняет преобразование логических значений к числовому виду данных). Например, рассмотрим этапы вычислений при использовании данного выражения в записи $=ЕСЛИ((5>2)*(2>7);"true";"false")$:

- =ЕСЛИ((ИСТИНА)*(ЛОЖЬ);"true";"false");
- =ЕСЛИ(1*0;"true";"false");
- =ЕСЛИ(0;"true";"false");
- =ЕСЛИ(ЛОЖЬ;"true";"false");
- Возвращен результат "false".

Примечания 3:

1. Функция ИЛИ чаще всего используется для расширения возможностей других функций.
2. В отличие от функции И, функция или эквивалентна операции сложения. Например, записи $=ИЛИ(10=2;5>0)$ и $=(10=2)+(5>0)$ вернут значение ИСТИНА и 1 соответственно, при этом значение 1 может быть интерпретировано как логическое ИСТИНА.
3. Подобно функции И, функция ИЛИ может быть использована как формула массива для множественной проверки условий.